/*
 *  Licensed to the Apache Software Foundation (ASF) under one
 *  or more contributor license agreements.  See the NOTICE file
 *  distributed with this work for additional information
 *  regarding copyright ownership.  The ASF licenses this file
 *  to you under the Apache License, Version 2.0 (the
 *  "License"); you may not use this file except in compliance
 *  with the License.  You may obtain a copy of the License at
 *
 *    http://www.apache.org/licenses/LICENSE-2.0
 *
 *  Unless required by applicable law or agreed to in writing,
 *  software distributed under the License is distributed on an
 *  "AS IS" BASIS, WITHOUT WARRANTIES OR CONDITIONS OF ANY
 *  KIND, either express or implied.  See the License for the
 *  specific language governing permissions and limitations
 *  under the License.
 */

if (sharedConfiguration.hasCodeCoverage.get()) {
    // we only apply the jacoco configuration if a system property is set, in order to avoid instrumentation if we
    // are not specifically asking for code coverage
    tasks.register("jacocoAllReport", JacocoReport) {
        allprojects { p ->
            p.plugins.withType(JacocoPlugin).configureEach {
                if (p.sourceSets.test.allSource.srcDirs.any { it.exists() }) {
                    def testTasks = []
                    p.tasks.matching({ t -> t.extensions.findByType(JacocoTaskExtension) }).each { testTask ->
                        if ((testTask instanceof Test)) {
                            sourceSets p.sourceSets.main
                            executionData(testTask)
                            testTasks << testTask
                        }
                    }

                    testTasks.each {
                        rootProject.tasks.jacocoAllReport.dependsOn(it)
                    }
                }
            }
        }

        reports {
            xml.required = true
            csv.required = true
        }
    }
}
